

A Guide to XLOOKUP in Excel

Mike Thomas

CANTOR *Pittgerald*

Introduction: What is XLOOKUP?	1
Search Down a Column	2
Search Across a Row	3
Optional Parameter: If_Not_Found	4
Optional Parameter: Match_Mode	5
Exact Match or Next Smaller Item	6
Exact Match or Next Larger Item	7
Wildcard Character Match	8
Optional Parameter: Search_Mode	9
Reverse Search	9
Binary Search	9
Return Multiple Values	10
Two-Way Lookup	11
About Mike Thomas	12

Introduction: What is XLOOKUP?

A common task performed on data that is stored in a tabular layout in Excel is to search down a column or across a row for a specific value (number, text, date) and then return a corresponding value from another column on the same row or another row in the same column.

The XLOOKUP function is designed to do this, saving you time and reducing the risk of errors.

The XLOOKUP function has 3 mandatory parameters and 3 optional parameters

The mandatory parameters are:

- The value that you are searching for (the Lookup Value)
- The range (single column or single row) that contains the value (the Lookup Array)
- The range (single column or single row) that contains the answer (the Result Array)

The optional parameters are:

- If_Not_Found
- Match_Mode
- Search_Mode

Both the mandatory and the optional parameters will be covered in detail in this document.

Search Down a Column

Imagine that you have been sent a list of mobile phone details along with the name and EmployeeID of the owner and asked to lookup the name of the owner and the model of the phone whose number matches the one in C3.

	A	B	C	D	E	F	G	H	I	J
1										
2		Phone Details								
3		Phone Number:	+44 7700 900161	TET-123	Harry	+44 7700 900584	iPhone 13 Pro Max	Silver		April-2022
4		Model:	iPhone 13 Pro	TET-321	John	+44 7700 900841	iPhone 12	Black		April-2022
5		Name:	Cath	TET-544	Adam	+44 7700 900401	iPhone 13 Pro	Gold		May-2022
6				TET-083	Mike	+44 7700 900880	iPhone 11 Pro	Space Grey		June-2022
7				TET-013	Cath	+44 7700 900161	iPhone 13 Pro	Silver		July-2022
8				TET-488	Graham	+44 7700 900094	iPhone 11	White		December-2022
9				TET-993	Kerry	+44 7700 900706	iPhone 12	Black		March-2023
10				TET-155	Paula	+44 7700 900702	iPhone 13 Pro	Gold		April-2023
11										

The formulas to do this are:

C4: =XLOOKUP(C3,G3:G10,H3:H10)

C5: = XLOOKUP(C3,G3:G10,F3:F10)

In other words...

- Treat what is in C3 as the Lookup Value.
- Look for this item in the range G3:G10
- When you find it, retrieve the value from column H from the same row (column F in the case of the formula in C5)

Search Across a Row

Here is a list of sports events and the names of the competitors in those events together with the finishing position of each person in each event. The requirement is to lookup the position of a specific person in a specific event.

The screenshot shows an Excel spreadsheet with the following data:

Name & Event Details	
Name:	Bill
Event:	3
Position:	1

The Great Excel Athletics Event								
Name	Adam	Alison	Bill	Cath	Graham	Kerry	Mike	Paula
Event 1	5	7	6	8	2	1	4	3
Event 2	8	5	3	7	6	1	2	2
Event 3	3	5	1	2	4	7	8	6
Event 4	5	3	4	7	8	2	1	6
Event 5	8	6	4	7	2	3	5	1
Event 6	2	3	7	1	4	8	6	5
Event 7	2	4	6	1	7	5	8	3

The formula bar at the top shows: `=XLOOKUP(C3,F3:M3,F6:M6)`

The formula in C5 is `=XLOOKUP(C3,F3:M3,F6:M6)`

Excel interprets this as...


- Treat what is in C3 as the Lookup Value
- Look for this value in the range F3:M3
- When you find the value, retrieve the contents of row 6 from the same column (row 6 contains the finishing positions for Event 3)

In the real world, the 3rd parameter would need to be changed depending on the event, but the example serves to demonstrate the principle.

Optional Parameter: If_Not_Found

There may be situations where Excel is unable to find the Lookup Value in the Lookup Array. When that happens, the XLOOKUP function returns #N/A as the result.

In the example below, because the phone number entered in C3 does not exist in G3:G10, the result of the formula is displayed as #N/A.



The screenshot shows an Excel spreadsheet with a table of phone details and a lookup result. The table has columns: ID, Name, Number, Model, Colour, and Contract Renewal. The lookup result is displayed as #N/A in cell C4.

ID	Name	Number	Model	Colour	Contract Renewal
TET-123	Harry	+44 7700 900584	iPhone 13 Pro Max	Silver	April-2022
TET-321	John	+44 7700 900841	iPhone 12	Black	April-2022
TET-544	Adam	+44 7700 900401	iPhone 13 Pro	Gold	May-2022
TET-083	Mike	+44 7700 900880	iPhone 11 Pro	Space Grey	June-2022
TET-013	Cath	+44 7700 900161	iPhone 13 Pro	Silver	July-2022
TET-488	Graham	+44 7700 900094	iPhone 11	White	December-2022
TET-993	Kerry	+44 7700 900706	iPhone 12	Black	March-2023
TET-155	Paula	+44 7700 900702	iPhone 13 Pro	Gold	April-2023

Although displaying #N/A alerts us to the fact that there is a problem, it looks untidy.


The traditional way to resolve this is to wrap the XLOOKUP function inside an IFERROR function, for example: **=IFERROR(XLOOKUP(C3,G3:G10,H3:H10),"Invalid Phone Number")**

This will display the text *Invalid Phone Number* in C4 if the XLOOKUP function results in #N/A (Excel treats #N/A as an error)

A simpler way to achieve the same result is to include "Invalid Phone Number" as the 4th parameter of the XLOOKUP function:

=XLOOKUP(C3,G3:G10,H3:H10, "Invalid Phone Number")

If the item in C3 cannot be found in G3:G10, display *Invalid Phone Number* as the formula result.



The screenshot shows an Excel spreadsheet with a table of phone details and a lookup result. The table has columns: ID, Name, Number, Model, Colour, and Contract Renewal. The lookup result is displayed as Invalid Phone Number in cell C4.

ID	Name	Number	Model	Colour	Contract Renewal
TET-123	Harry	+44 7700 900584	iPhone 13 Pro Max	Silver	April-2022
TET-321	John	+44 7700 900841	iPhone 12	Black	April-2022
TET-544	Adam	+44 7700 900401	iPhone 13 Pro	Gold	May-2022
TET-083	Mike	+44 7700 900880	iPhone 11 Pro	Space Grey	June-2022
TET-013	Cath	+44 7700 900161	iPhone 13 Pro	Silver	July-2022
TET-488	Graham	+44 7700 900094	iPhone 11	White	December-2022
TET-993	Kerry	+44 7700 900706	iPhone 12	Black	March-2023
TET-155	Paula	+44 7700 900702	iPhone 13 Pro	Gold	April-2023

Optional Parameter: Match_Mode

There may be situations where, if an exact match cannot be found, you still want Excel to return a value from the Result Array instead of #N/A or the value of the If_Not_Found parameter if it has been defined.

To do this you must define a value for the Match_Mode parameter. This parameter is the 5th parameter of the XLOOKUP function. Select one of the following 4 choices:


- 0: Exact Match (the default if the parameter is omitted)
- 1: Exact Match or Next Smaller Item
- 1: Exact Match or Next Larger Item
- 2: Wildcard Character Match

Each of these options is explained in detail in the following pages.

Exact Match or Next Smaller Item

Here is a list of dates and seasons. The requirement is to specify, in C4, the season, based on the date in C3. The formula in C4 is **=XLOOKUP(C3,F3:F6,E3:E6)**

As no Match_Mode parameter has been defined, Excel searches for an exact match but the date in C3 does not match exactly one of the dates in F3:F6 therefore the formula returns #N/A




The screenshot shows an Excel spreadsheet with the following data:

Transaction Details		Season	Season Start
Transaction Date:	12 August 2021	Winter	21 December 2021
Season:	#N/A	Spring	20 March 2021
		Summer	20 June 2021
		Autumn / Fall	22 September 2021

The formula bar for cell C4 shows: **=XLOOKUP(C3,F3:F6,E3:E6)**

By amending the formula to **=XLOOKUP(C3,F3:F6,E3:E6,-1)**, if an exact match cannot be found, Excel looks for "the next earliest date" (i.e. next smallest item)



The screenshot shows the same Excel spreadsheet as before, but with the formula in C4 amended to **=XLOOKUP(C3,F3:F6,E3:E6,-1)**. The result in C4 is now **Summer**.

Transaction Details		Season	Season Start
Transaction Date:	12 August 2021	Winter	21 December 2021
Season:	Summer	Spring	20 March 2021
		Summer	20 June 2021
		Autumn / Fall	22 September 2021

The formula bar for cell C4 shows: **=XLOOKUP(C3,F3:F6,E3:E6,-1)**

In other words, 12th August does not exist in F3:F6 so Excel searches for the first date BEFORE that date, which in this example is 20 June.

Note that it doesn't matter whether the list is sorted into date order or not


IMPORTANT: The If_Not_Found parameter has deliberately not been defined because a match (exact or otherwise) will always be found. However, because the Match_Mode parameter is the 5th parameter and If_Not_Found is the 4th parameter, it is necessary to include the comma separator otherwise Excel would treat -1 as the If_Not_Found parameter. It is not necessary to put anything between the commas.

If the Lookup Value is numeric (i.e. not date or text), setting the Match_Mode parameter to -1 will find the next smallest number, for example, suppose the Lookup Value is 10 and the values in the Lookup Array are 3,5,7,9,11,13,15. The result of the formula will be 9.

Exact Match or Next Larger Item


The worksheet below is used by warehouse packers at a company that sells bottles of wine. The packers use the worksheet to determine how many "packing boxes" are required. Each box can hold 12 bottles. In this example a customer has ordered 18 bottles of wine.

Omitting the Match_Mode parameter will result in #N/A if the Quantity Ordered (C3) does not match any of the values in F3:F7.



Order Details		No of Bottles	No of Boxes
Quantity Ordered:	18	12	1
Boxes Needed:	#N/A	24	2
		36	3
		48	4
		60	5


Specifying -1 as the Match_Mode parameter (next smaller item) will result in not enough packing boxes being requested by the packing team...



Order Details		No of Bottles	No of Boxes
Quantity Ordered:	18	12	1
Boxes Needed:	1	24	2
		36	3
		48	4
		60	5

In this particular example, if the quantity ordered (C3) does not match exactly one of the values in F3:F7, the next highest value from column F is needed. In other words, 18 bottles require 2 packing boxes - one to hold 12 bottles and one to hold the remaining 6 bottles.

In F3:F7, 18 is between 12 and 24 so 24 is the "next larger item" and 2 is the value in the Return Array for 24. The correct formula for B2 is: **=XLOOKUP(C3,F3:F7,G3:G7,,1)**



Order Details		No of Bottles	No of Boxes
Quantity Ordered:	18	12	1
Boxes Needed:	2	24	2
		36	3
		48	4
		60	5

Wildcard Character Match

To perform a partial match (also known as a wildcard match), set the value of the Match_Mode parameter to 2.

The wildcard characters can either be typed into the cell containing the Lookup Value or can be included as part of the XLOOKUP formula.

In this example, the owner of the wine store wants to allow the staff to type in part of the wine name into C3 and return the price of that wine in C4.

Price Query		Name	Price
Wine Name:	Ch	Prosecco	£8.50
Price:	£6.00	Sauvignon Blanc	£9.00
		Chardonnay	£6.00
		Shiraz	£8.00
		Merlot	£6.25

The formula in C4 is: =XLOOKUP("*" & C3 & "*",F3:F7,G3:G7,"Unknown Wine",2)

The * character is a wildcard that represents any number of characters, so...

CH* means anything that begins with CH

*CH means anything that ends with CH

CH means anything that contains CH

What happens if you don't specify 2 as the Match_Mode parameter? Excel searches for the literal string *CH*

Price Query		Name	Price
Wine Name:	Ch	Prosecco	£8.50
Price:	Unknown Wine	Sauvignon Blanc	£9.00
		Chardonnay	£6.00
		Shiraz	£8.00
		Merlot	£6.25

Optional Parameter: Search_Mode

By default, Excel retrieves the value from the Result Array based on the FIRST occurrence of the Lookup Value in the Lookup Array.

Reverse Search

However, by specifying -1 as the 6th parameter (known as the Search_Mode parameter), Excel will perform a "reverse search", i.e. search upwards from the bottom of the list for a vertical lookup and search leftwards from the right hand side for a horizontal search.

The formula in C4 is: =XLOOKUP(C3,F3:F10,G3:G10,"Unknown Name",-1)

C4 ∨									
---	--	--	--	--	--	--	--	--	--

Excel searches for the value in C3 (Cath) upwards starting at F10 rather than downwards starting at F3. The first occurrence of Cath is in F7.

Binary Search

Setting the Search_Mode parameter to 2 or -2 will perform a binary search which, when the XLOOKUP function was first launched, resulted in a faster search. However, this is no longer the case since Microsoft made improvements to the lookup algorithms. In other words, it is unlikely that you would ever need to set the 6th parameter to anything other than -1.

If you omit the 6th parameter, it will default to 1 which is a top-to-bottom or left-to-right search.

Return Multiple Values

By setting the Return Array parameter to a multi-column range, the XLOOKUP function can retrieve an entire row or column of data related to the Lookup Value.

Here is a list of sports events and the names of the competitors in those events together with the finishing position of each person in each event. The requirement is to retrieve the positions of a specific person in all the events that they participated in.

The screenshot shows an Excel spreadsheet with two tables. The first table, located in the range B3:I10, lists competitors and their finishing positions across seven events. The second table, located in the range K2:R8, shows the result of an XLOOKUP formula for the competitor 'Graham', returning their positions in all seven events. The formula bar at the top shows the formula: =XLOOKUP(K2,B3:B10,C3:I10).

Name	Event 1	Event 2	Event 3	Event 4	Event 5	Event 6	Event 7
Adam	5	8	3	5	8	2	2
Alison	7	5	5	3	6	3	4
Bill	6	3	1	4	4	7	6
Cath	8	7	2	7	7	1	1
Graham	2	6	4	8	2	4	7
Kerry	1	1	7	2	3	8	5
Mike	4	4	8	1	5	6	8
Paula	3	2	6	6	1	5	3

Graham	Event 1	Event 2	Event 3	Event 4	Event 5	Event 6	Event 7
	2	6	4	8	2	4	7

The formula in L3 is =XLOOKUP(K2,B3:B10,C3:I10)

Notice that the 3rd parameter (the Return Array) is a multi-column range. Excel searches B3:B10 for the value in K2. When it finds it, it returns the values from columns C:I from the appropriate row.

Two-Way Lookup

Here is a list of sports events and the names of the competitors in those events together with the finishing position of each person in each event. The requirement is to lookup the position of a specific person in a specific event.

The traditional way of doing this is to use a combination of INDEX and MATCH functions however XLOOKUP provides a simpler way, using a Nested XLOOKUP. It takes advantage of being able to specify a multi-column-multi-row range as the Return Array

The formula in C5 is: =XLOOKUP(C3,E3:E10,XLOOKUP(C4,F2:L2,F3:L10))

	A	B	C	D	E	F	G	H	I	J	K	L	M
2		Name / Event Lookup			Name	Event 1	Event 2	Event 3	Event 4	Event 5	Event 6	Event 7	
3		Name:	Bill		Adam	5	8	3	5	8	2	2	
4		Event:	Event 3		Alison	7	5	5	3	6	3	4	
5		Position:	1		Bill	6	3	1	4	4	7	6	
6					Cath	8	7	2	7	7	1	1	
7					Graham	2	6	4	8	2	4	7	
8					Kerry	1	1	7	2	3	8	5	
9					Mike	4	4	8	1	5	6	8	
10					Paula	3	2	6	6	1	5	3	
11													
12													

The first XLOOKUP searches E3:E10 for the value that is in C3 (the competitor's name)

The Return Array is another XLOOKUP which searches F2:L2 for the value that is in C4 (the event name) and returns a value from the multi-column-multi-row range F3:L10 (the value returned is from the intersection of the specified Name and Event defined in C3 and C4)

About Mike Thomas

Mike Thomas is a subject matter expert in a range of technologies, although his primary focus and passion is Excel and Power BI.

In a career spanning 30+ years, he has delivered thousands of training courses on a wide variety of technology-related topics.

In 2012 Mike founded theexceltrainer.co.uk where he has produced nearly 200 written and video-based Excel tutorials. He also runs 2 YouTube Channels and has recorded several video-based training courses for pluralsight.com.

Mike is a Fellow of The Learning and Performance Institute and has worked with and for many global and UK-based companies and organizations across a diverse range of sectors.

In addition to training, he also designs and develops Microsoft Office-based solutions that automate key business tasks and processes.

Away from the Microsoft world, Mike is co-host of the [MacBites](#) podcast and a co-presenter of [MacBites After Hours](#), a live technology show hosted on YouTube.

Connect with Mike via the links below

- [YouTube Channel 1](#)
- [YouTube Channel 2](#)
- [The Excel Trainer Website](#)
- [LinkedIn](#)